Mitigation of rocuronium induced residual neuromuscular blockade risk by means of a population PK/PD model

Miriam Happ[1,2], Leandro Pippa[2], Gabriela Lauretti[3], Anthony Gebhart[2], Günther Weindl[1], Francine Azeredo[2], Valvanera Vozmediano[2], Stephan Schmidt[2], Natalia de Moraes[2]

[1]Pharmacology and Toxicology Section, Pharmaceutical Institute, University of Bonn, Germany. [2]Center for Pharmacometrics and System Pharmacology, Department of Pharmaceutics. College of Pharmacy. University of Florida, FL, USA. [3]School of Medicine of Ribeirao Preto, University of Sao Paulo, SP, Brazil

Supporting material

Derivation of dual input effect compartment

$$V_e \frac{dCe}{dt} = k_{1e} \times Cc \times V1 + k_{2e} \times Cp \times V2 - k_{e0} \times Ce \times V_e \quad (1)$$

$$\frac{dCe}{dt} = \frac{V_1}{V_e} k_{1e} \times Cc + \frac{V_2}{V_e} k_{2e} \times \mathrm{Cp} - k_{e0} \times Ce \quad (2)$$

at equilibrium $V_e \frac{dCe}{dt} = 0$ and $Cc = Cp = Ce$

$$V_1 \times k_{1e} + V_2 \times k_{2e} = V_e \times k_{e0} \quad (3)$$

$$V_e = \frac{V_1 \times k_{1e} + V_2 \times k_{2e}}{k_{e0}} \quad (4)$$

$$\frac{dCe}{dt} = k_{e0} \times \left(\frac{V_1 \times k_{1e} \times Cc + V_2 \times k_{2e} \times Cp}{V_1 \times k_{1e} + V_2 \times k_{2e}} - Ce \right) \quad (4) \text{ in } (2) \rightarrow (5)$$

if $k_{1e} = k_{2e}$:

$$\frac{dCe}{dt} = -k_{e0} \times \left(Ce - \frac{V_1 \times Cc + V_2 \times Cp}{V_1 + V_2} \right) \quad (6)$$

Mlxtran code for structural pharmacokinetic/pharmacodynamic model with two pharmacokinetic compartments, linear elimination from the central compartment, and a sigmoid I max model linked to the central and peripheral compartment via a dual input effect compartment.

```
DESCRIPTION:
Administration: IV bolus into the central compartment.
PK-model has two compartments: a central compartment (volume V), a peripheral compartment (transfer rate to and from k12 and k21), and a linear elimination from the central compartment (elimination rate k).
The PD model is an I-max model with an effect compartment and a full inhibition (Imax = 1) at high concentrations. The model has a baseline effect E0 and a half maximal inhibitory concentration IC50. The effect compartment (elimination rate ke0) is linked to the central and peripheral compartments. The parameter gamma accounts for the sigmoidicity of the drug effect.

[LONGITUDINAL]
input = {V, k, k12, k21, ke0, gamma, E0, IC50}

PK:
; parameter transformations
V1 = V
Cl = k*V1
Q  = k12*V1
V2 = Q/k21

; administration
depot (type = 1, target = Ac)

EQUATION:
; initial conditions
t_0  = 0
Ac_0 = 0
Ap_0 = 0
Ce_0 = 0
; PK compartments
ddt_Ac = k21*Ap - k12*Ac - k*Ac
Cc = Ac/V1

ddt_Ap = k12* Ac - k21*Ap
Cp = Ap/V2

; effect compartment
ddt_Ce = - ke0 * (Ce - ((V*Cc + V2*Cp)/(V + V2)))

; concentration-effect relationship
E = E0 * (1 - max(Ce,0) ^ gamma / (max(Ce,0) ^ gamma + IC50 ^ gamma))

OUTPUT:
output = {Cc, E, Ce}
```



Figure 1: Concentration-time profiles of rocuronium for all subjects in semi-logarithmic scaling (A); train of four ratio versus rocuronium plasma concentration (B); and train of four ratio over time (C). The investigated subjects received single intravenous bolus doses of 0.3–1.2 mg/kg rocuronium.